# $Id: PKGBUILD 70305 2012-05-04 08:21:08Z mtorromeo $
# Maintainer: Massimiliano Torromeo <massimiliano.torromeo@gmail.com>
# Maintainer: Felix Yan <felixonmars@gmail.com>
# Contributor: Ralf Schmitt <ralf@systemexit.de>

pkgname=python2-gevent
pkgver=0.13.8
pkgrel=2
arch=('i686' 'x86_64')
pkgdesc="Python network library that uses greenlet and libevent for easy and scalable concurrency"
license=("MIT")
url="http://www.gevent.org/"
depends=('libevent' 'python2-greenlet')
makedepends=('python2-distribute')

source=(http://pypi.python.org/packages/source/g/gevent/gevent-$pkgver.tar.gz)

build() {
	cd "$srcdir/gevent-$pkgver"
	python2 setup.py build
}

package() {
	cd "$srcdir/gevent-$pkgver"
	python2 setup.py install -O1 --root="$pkgdir"
	install -Dm0644 LICENSE "$pkgdir/usr/share/licenses/$pkgname/LICENSE"
	install -Dm0644 LICENSE.pyevent "$pkgdir/usr/share/licenses/$pkgname/LICENSE.pyevent"
}

md5sums=('ca9dcaa7880762d8ebbc266b11252960')
